```
> restart:
```

Leçon 5 : Intégration basique.

```
> int(1/(1+x^2),x=0..1);
```

$$\frac{1}{4}\,\pi$$

```
> int(1/(1+x^2),x);
```

$$\arctan(x)$$

De toutes façons vous finirez par appeler une intégrale I, qui est l'imaginaire pur bien connu, et vous aurez droit à ceci !

```
> I := int(1/(1+x^2),x);
Error, Illegal use of an object as a name
```

Maple accepte les intervalles non bornés.

```
> int(exp(-x^2/2),x=-infinity..infinity);
```

$$\sqrt{2}\,\sqrt{\pi}$$

Maple possède une bibliothèque de fonctions qui s'expriment à l'aide d'intégrales...et ne manque pas de s'en servir (faire help Ei pour plus de détails...)

```
> int(exp(x)/x,x);
```

$$-\mathrm{Ei}(1,\,-x)$$

```
> int(exp(x)/x,x=1..2); evalf(");
```

$$-\mathrm{Ei}(1,\,-2)+\mathrm{Ei}(1,\,-1)$$

$$3.059116540$$

```
> f := 1/(x^4+x^2+1) : int(f,x);
```

$$\frac{1}{4}\ln(x^2+x+1)+\frac{1}{6}\sqrt{3}\arctan(\frac{1}{3}\,(2\,x+1)\,\sqrt{3})-\frac{1}{4}\ln(x^2-x+1)+\frac{1}{6}\sqrt{3}\arctan(\frac{1}{3}\,(2\,x-1)\,\sqrt{3})$$

A titre de vérification, voici comment obtenir la décomposition en éléments simples (ce qui n'est pas évident à dénicher dans la documentation)

```
> convert(f,parfrac,x);
```

$$-\frac{1}{2}\,\frac{-1+x}{x^2-x+1}+\frac{1}{2}\,\frac{1+x}{x^2+x+1}$$

```
> int(",x);
```

$$\frac{1}{4}\ln(x^2+x+1)+\frac{1}{6}\sqrt{3}\arctan(\frac{1}{3}\,(2\,x+1)\,\sqrt{3})-\frac{1}{4}\ln(x^2-x+1)+\frac{1}{6}\sqrt{3}\arctan(\frac{1}{3}\,(2\,x-1)\,\sqrt{3})$$

Décomposer sur C est plus acrobatique, si on ne connait pas les facteurs : le quatrième paramètre indique que les facteurs doivent être de la forme a+ib*sqrt(3). Il ne semble pas que Maple factorise de lui-même sur C, sauf numeriquement.

```
> convert(f,parfrac,x,(-3)^(1/2));convert(f,par
> frac,x,complex);
```

$$-\frac{1}{6}\frac{-3+I\sqrt{3}}{2x+1-I\sqrt{3}}+\frac{1}{6}\frac{3+I\sqrt{3}}{2x+1+I\sqrt{3}}+\frac{1}{6}\frac{-3+I\sqrt{3}}{2x-1+I\sqrt{3}}-\frac{1}{6}\frac{3+I\sqrt{3}}{2x-1-I\sqrt{3}}$$

$$\frac{.2499999999+.1443375672\,I}{x+.5000000000+.8660254038\,I}+\frac{.2499999999-.1443375672\,I}{x+.5000000000-.8660254038\,I}+\frac{-.2499999999+.1443375672\,I}{x-.5000000000+.8660254038\,I}$$
$$+\frac{-.2499999999-.1443375672\,I}{x-.5000000000-.8660254038\,I}$$

Il n'y a pas réellement d'intégrales doubles dans Maple, mais on peut "Fubiniser".

```
> f := x^2+x*y:int(int(f,y=0..x),x=0..1);
```

$$\frac{3}{8}$$

Maple sait calculer symboliquement avec les intégrales

```
> an := int(phi(x)*cos(n*x),x=-Pi..Pi)/Pi;
```

$$an := \frac{\int_{-\pi}^{\pi}\phi(x)\cos(n\,x)\,dx}{\pi}$$

```
> phi := x->abs(x):an;
```

$$2\,\frac{\cos(\pi\,n)-1+n\sin(\pi\,n)\,\pi}{n^2\,\pi}$$

```
> assume(n, integer); an;
```

$$2\,\frac{(-1)^{n\tilde{}}-1}{n\tilde{}^{\,2}\,\pi}$$

Forme inerte, integration par parties, changement de variable.

Maple permet de définir une forme *inerte* (ou *symbolique*) pour l'intégrale, qui correspond vaguement à l'idée de primitive, i.e. une expression non évaluée directement mais ayant certaines propriétés...

```
> I1 :=
> Int(1/(1+x^2),x);diff(I1,x);series(I1,x,8);
```

$$I1 := \int\frac{1}{1+x^2}\,dx$$

$$\frac{1}{1+x^2}$$

$$x-\frac{1}{3}x^3+\frac{1}{5}x^5-\frac{1}{7}x^7+\mathrm{O}(x^9)$$

I1 est une integrale "symbolique", que l'on peut néanmoins évaluer, si cela est possible, non
pas avec l'instruction eval mais en utilisant l'instruction value .

```
> eval(I1);
```

$$\int \frac{1}{1+x^2}\,dx$$

```
> value(I1);
```

$$\arctan(x)$$

Les calculs "classiques" sont définis, entre autres, dans le package student, mais ne s'appliquent généralement qu'à la forme inerte.

```
> with(student);
```

$[D, \mathit{Diff}, \mathit{Doubleint}, \mathit{Int}, \mathit{Limit}, \mathit{Lineint}, \mathit{Product}, \mathit{Sum}, \mathit{Tripleint}, \mathit{changevar}, \mathit{completesquare}, \mathit{distance}, \mathit{equate}, \mathit{integrand}, \mathit{intercept}, \mathit{intparts}, \mathit{leftbox}, \mathit{leftsum}, \mathit{makeproc}, \mathit{middlebox}, \mathit{middlesum}, \mathit{midpoint}, \mathit{powsubs}, \mathit{rightbox}, \mathit{rightsum}, \mathit{showtangent}, \mathit{simpson}, \mathit{slope}, \mathit{summand}, \mathit{trapezoid}]$

```
> I2 := Int(t/(1+t^2)/sqrt(1-t^4),t=0..1);
```

$$\mathit{I2} := \int_0^1 \frac{t}{(1+t^2)\sqrt{1-t^4}}\,dt$$

```
> value(I2);
```

$$\frac{1}{2} + \left(\sum_{\alpha=\%1} (\frac{1}{8}\sqrt{2}\,\_\alpha\,\mathrm{EllipticPi}(\frac{1}{2}, \frac{1}{2}\sqrt{2}))\right)$$
$$\%1 := \mathrm{RootOf}(\_Z^2+1)$$

Intéressant, mais peu pratique : cela doit pouvoir s'arranger...

```
> evalf(");
```

$$.5000000000$$

On va donc mâcher le travail de Maple.

```
> I3 := changevar(u=t^2,I2,u);
```

$$\mathit{I3} := \int_0^1 \frac{1}{2}\,\frac{1}{(1+u)\sqrt{1-u^2}}\,du$$

```
> I4 := changevar(u=cos(x),I3,x);
```

$$\mathit{I4} := \int_0^{1/2\,\pi} \frac{1}{2}\,\frac{\sin(x)}{(1+\cos(x))\sqrt{1-\cos(x)^2}}\,dx$$

C'est là que cela se gâte...

```
> I5 := simplify(I4,symbolic);
```

$$\mathit{I5} := \frac{1}{2}\int_0^{1/2\,\pi} \frac{1}{1+\cos(x)}\,dx$$

```
> value(");
```

$$\frac{1}{2}$$

On peut remarquer que le fait d'avoir une intégrale impropre ne gêne aucunement Maple.

```
> f := 1/(1+x^2)^n:In := Int(f,x);
```

$$In := \int \frac{1}{(1+x^2)^{n\text{~}}}\,dx$$

```
> Jn:= intparts(In,f);
```

$$Jn := \frac{x}{(1+x^2)^{n\text{~}}} - \int -2\,\frac{n\text{~}\,x^2}{(1+x^2)^{n\text{~}}\,(1+x^2)}\,dx$$

```
> simplify(Jn);
```

$$(1+x^2)^{(-n\text{~})}\,x + 2\,n\text{~}\int (1+x^2)^{(-n\text{~}-1)}\,x^2\,dx$$

```
> restart:
```

Séries etc...

Diverses manières de travailler avec les séries, applications aux séries de Fourier.
D'abord les séries de "Taylor".

```
> f1 := sqrt(1+x+x^2);
```

$$f1 := \sqrt{1+x+x^2}$$

```
> series(f1,x=0,4);
```
> le dernier paramètre est l'ordre désiré.

$$1 + \frac{1}{2}\,x + \frac{3}{8}\,x^2 - \frac{3}{16}\,x^3 + \frac{3}{128}\,x^4 + \frac{15}{256}\,x^5 + \mathrm{O}(x^6)$$

```
> series(f1,x=0,8);
```

$$1 + \frac{1}{2}\,x + \frac{3}{8}\,x^2 - \frac{3}{16}\,x^3 + \frac{3}{128}\,x^4 + \frac{15}{256}\,x^5 - \frac{57}{1024}\,x^6 + \frac{21}{2048}\,x^7 + \frac{867}{32768}\,x^8 - \frac{1893}{65536}\,x^9 + \mathrm{O}(x^{10})$$

```
> Order := 6:
```
> on peut aussi utiliser la variable Order

```
> ee := series(exp(x),x);
```
> x sans indication sgnifie x=0

$$ee := 1 + x + \frac{1}{2}\,x^2 + \frac{1}{6}\,x^3 + \frac{1}{24}\,x^4 + \frac{1}{120}\,x^5 + \mathrm{O}(x^6)$$

```
> series(n!,n=infinity,3);
```

$$\frac{\dfrac{\sqrt{2}\,\sqrt{\pi}}{\sqrt{\dfrac{1}{n}}} + \dfrac{1}{12}\,\sqrt{2}\,\sqrt{\pi}\,\sqrt{\dfrac{1}{n}} + \dfrac{1}{288}\,\sqrt{2}\,\sqrt{\pi}\,(\dfrac{1}{n})^{3/2} + \mathrm{O}((\dfrac{1}{n})^{5/2})}{(\dfrac{1}{n})^n\,e^n}$$

Maple connait le formule de Stirling...et sait faire des developpements génralisés !

```
>  p := convert(ee,polynom);
>  pour récuperer la partie régulière
```

$$p := 1 + x + \frac{1}{2}\,x^2 + \frac{1}{6}\,x^3 + \frac{1}{24}\,x^4 + \frac{1}{120}\,x^5$$

```
>  solve(ee=y,x); #
>  on peut même calculer le developpement d'une fonction réciproque ... y doit être une variable libre !
```

$$y - 1 - \frac{1}{2}\,(y-1)^2 + \frac{1}{3}\,(y-1)^3 - \frac{1}{4}\,(y-1)^4 + \frac{1}{5}\,(y-1)^5 + \mathrm{O}((y-1)^6)$$

add est utilisé pour faire des sommations (plus efficaces que les boucles ...)

```
>  add(1/k,k=1..30);
```

$$\frac{9304682830147}{2329089562800}$$

```
>  add(x^k/k!,k=1..10);
```

$$x + \frac{1}{2}\,x^2 + \frac{1}{6}\,x^3 + \frac{1}{24}\,x^4 + \frac{1}{120}\,x^5 + \frac{1}{720}\,x^6 + \frac{1}{5040}\,x^7 + \frac{1}{40320}\,x^8 + \frac{1}{362880}\,x^9 + \frac{1}{3628800}\,x^{10}$$

```
>  add(k^2,k=1..n);
```

```
Error, unable to execute add
```

sum est utilisé pour faire des sommes formelles ...

```
>  sum(k^2,k=1..n);
```

$$\frac{1}{3}\,(n+1)^3 - \frac{1}{2}\,(n+1)^2 + \frac{1}{6}\,n + \frac{1}{6}$$

```
>  sum(x^k/k!,k=1..infinity);
>  et même avec une infinité de termes.
```

$$e^x\,(1 - e^{(-x)})$$

On peut aussi étudier la convergence des séries :

```
>  sum((-1)^n/((-1)^n+sqrt(n)), n=2..infinity);#
>  Maple sèche...mais pas nous!
```

$$\sum_{n=2}^{\infty} \frac{(-1)^n}{(-1)^n + \sqrt{n}}$$

```
>  DA:=asympt(epsilon/(epsilon+sqrt(n)),n,2);#
>  donc elle diverge
```

$$DA := \varepsilon\,\sqrt{\frac{1}{n}} - \frac{\varepsilon^2}{n} + \varepsilon^3\,(\frac{1}{n})^{3/2} + \mathrm{O}(\frac{1}{n^2})$$

```
>  sum((-1)^n/ln((-1)^n+n),n=2..infinity):DA:=as
>  ympt(epsilon/ln(epsilon+n),n,2);# donc elle converge
```

$$DA := \frac{\varepsilon}{\ln(n)} - \frac{\varepsilon^2}{\ln(n)^2\,n} + \mathrm{O}(\frac{1}{n^2})$$

On peut aussi utiliser Maple et pour regarder quel type de singularité on a (fonctions holomorphes)

```
> series(1/sin(x), x, 5);type(", laurent);
```

$$x^{-1} + \frac{1}{6}\,x + \mathrm{O}(x^3)$$

$$\mathit{true}$$

```
> series(ln(x+x^2), x, 3);type(", laurent);
```

$$\ln(x) + x - \frac{1}{2}\,x^2 + \mathrm{O}(x^3)$$

$$\mathit{false}$$

Et aussi récupérer des coefficients du développement en série et donc le résidu .... :

```
> b := series(tan(x)/(x*sin(x)), x=0, 13);
```

$$b := x^{-1} + \frac{1}{2}\,x + \frac{5}{24}\,x^3 + \frac{61}{720}\,x^5 + \frac{277}{8064}\,x^7 + \frac{50521}{3628800}\,x^9 + \mathrm{O}(x^{11})$$

```
> op(b);#op(2i-1,b) coefficient de x à la
> puissance op(2i,b)
```

$$1,\ -1,\ \frac{1}{2},\ 1,\ \frac{5}{24},\ 3,\ \frac{61}{720},\ 5,\ \frac{277}{8064},\ 7,\ \frac{50521}{3628800},\ 9,\ \mathrm{O}(1),\ 11$$

Applications aux séries de Fourier.

```
> restart:
```

Définir des fonctions a(n) et b(n) correspondant à

```
> a := n -> int(f(t)*cos(n*t)/Pi,t=-Pi..Pi):
> b := n -> int(f(t)*sin(n*t)/Pi,t=-Pi..Pi):
> f := t -> abs(t);
```

$$f := \mathrm{abs}$$

Définir une fonction S : (n,x) —-> a0/2+somme des a(n)cos(nx)+b(n)sin(nx).

```
Calculer S(3,x), S(5,x).

Faire afficher successivement les tracés de S(1,x), S(2,x), S(5,x),
S(10,x) et S(20,x) sur
```

l'intervalle [-Pi,Pi] ainsi que celui de f.

On peut même visualiser la convergence uniforme grâce à l'animation des différentes représentations graphiques et à la commande display (séquence de graphiques,insequence=true) de la bibliothèque plots. On verra plus tard qu'il existe aussi une commande animate plus adaptée à des fonctions dont les deux paramètres sont réels.

```
> with(plots):
```

```
En déduire la valeur de la somme des inverses des carrés des entiers
impairs.
```

Un autre exemple ... Calculer les coefficients de Fourier de la fonction (2Pi-périodique) impaire, égale à 1 sur]0 ;Pi[; et définir S(n,x) comme précédemment.

```
> restart:
```

L'objet de l'exercice est de montrer graphiquement, puis par le calcul (Maple calcule !) que la convergence n'est pas uniforme.

```
> plot(S(10,x),x=0..Pi):
```

L'instruction suivante calcule la dérivée de S par rapport à x.

```
> ds:= D[2](S):
```

Expliquer la suite des instructions.

```
> M := n -> S(n,Pi/2/n):
> value(M(10));
```

$$4\,\frac{\frac{20}{19}\sin(\frac{1}{20}\,\pi)+\frac{20}{51}\sin(\frac{3}{20}\,\pi)+\frac{2}{15}\sqrt{2}+\frac{20}{91}\sin(\frac{7}{20}\,\pi)+\frac{20}{99}\sin(\frac{9}{20}\,\pi)}{\pi}$$

```
> evalf(");
```

$$1.179814019$$

```
> value(M(20));
```

$$4(\frac{40}{39}\sin(\frac{1}{40}\,\pi)+\frac{40}{111}\sin(\frac{3}{40}\,\pi)+\frac{4}{35}\sqrt{2-\sqrt{2}}+\frac{40}{231}\sin(\frac{7}{40}\,\pi)+\frac{40}{279}\sin(\frac{9}{40}\,\pi)+\frac{40}{319}\sin(\frac{11}{40}\,\pi)+\frac{40}{351}\sin(\frac{13}{40}\,\pi)$$
$$+\frac{4}{75}\sqrt{2+\sqrt{2}}+\frac{40}{391}\sin(\frac{17}{40}\,\pi)+\frac{40}{399}\sin(\frac{19}{40}\,\pi))\,/\pi$$

```
> evalf(");
```

$$1.179188137$$

```
> evalf(value(M(30)));
```

$$1.179072349$$

```
> # limit(value(M(n)),n=infinity);#
```

> plantage ! on va ruser ...

```
> with(student):
> phi := x -> 2*sin(x)/Pi/x;
```

$$\phi := x \to 2\,\frac{\sin(x)}{\pi\, x}$$

```
> LL := middlesum(phi(x),x=0..Pi,n);
```

> sommes de Riemann a point milieu

$$LL := \frac{\pi\left(\displaystyle\sum_{i=0}^{n-1}\left(2\,\frac{\sin\left(\frac{(i+\frac{1}{2})\,\pi}{n}\right)\,n}{\pi^2\,(i+\frac{1}{2})}\right)\right)}{n}$$

```
> M(n);
```

$$4\,\frac{\displaystyle\sum_{p=0}^{n-1}\frac{\sin(\frac{1}{2}\,\frac{(2\,p+1)\,\pi}{n})}{2\,p+1}}{\pi}$$

```
> LL -M(n);
```

$$\frac{\pi\left(\displaystyle\sum_{i=0}^{n-1}\left(2\,\frac{\sin\left(\dfrac{(i+\frac{1}{2})\,\pi}{n}\right)\,n}{\pi^2\,(i+\frac{1}{2})}\right)\right)}{n}-4\,\frac{\displaystyle\sum_{p=0}^{n-1}\frac{\sin(\frac{1}{2}\,\frac{(2\,p+1)\,\pi}{n})}{2\,p+1}}{\pi}$$

```
> simplify(");
```

> toute tentative de faire ecrire 0 est vaine ! Inutile de passer à la limite !

$$4\,\frac{\left(\displaystyle\sum_{i=0}^{n-1}\frac{\sin(\frac{1}{2}\,\frac{(2\,i+1)\,\pi}{n})}{2\,i+1}\right)-\left(\displaystyle\sum_{p=0}^{n-1}\frac{\sin(\frac{1}{2}\,\frac{(2\,p+1)\,\pi}{n})}{2\,p+1}\right)}{\pi}$$

```
> int(phi(x),x=0..Pi);
```

$$2\,\frac{\mathrm{Si}(\pi)}{\pi}$$

```
> evalf(");
```

$$1.178979744$$

Il n'y a donc pas convergence uniforme de la série de Fourier (Phénomène de Gibbs).